

MEMORY – Memory Game and Trainer

© 2019 Valentín Albillo

Abstract

MEMORY is a game program written in 1980 for the HP-41C programmable calculator to challenge your memory by testing your ability to remember what you've just seen and offering afterwards an accurate comment on your performance.

Keywords: Memory, game program, trainer, programmable calculator, RPN, HP-41C, HP-41CV, HP-41CX, HP42S

1. Introduction

MEMORY is a fun game program I wrote in 1980 for the HP-41C programmable calculator (will also run *as-is* in the HP-41CV/CX and the HP42S), to try and test your ability to accurately remember series of digits, shown as a variable-length integer number randomly generated by the program. Each number is displayed for about 2 seconds (a trifle more, actually), and then you're prompted to enter it. Your guess is scored depending on the number of correct digits and a total score is kept. Perfect recall will be rewarded with a longer number in the next round (up to 10 digits) while an incorrect guess will result in a shorter number instead (at least 1 digit).

You must enter at least one non-zero digit for every guess, ≤ 0 is not acceptable and will ask again to enter a guess. Some remarks:

- extra digits will be ignored: if 345 is correct then a guess of 345678 is totally right (scores 3 digits).
- wrong digits don't spoil all: if 269 is correct then guessing 26 scores 2 correct digits, and if 387 is correct then guessing 389 also scores 2 correct digits.

After 3 incorrect guesses the game ends and the score will be displayed, which will be the total number of digits correctly guessed, as well as a comment evaluating the quality of your memory based on the score just achieved.

The evaluation is based upon the score and the possible evaluations and their descriptions are summarized in the following table:

# Points	Evaluation	Description
< 15	NEAR AMNESIA	You should really seek help ASAP, memory this bad can be dangerous
15-29	SO PATHETIC	Your memory is seriously faulty or you need to pay more attention
30-44	VERY POOR	The evaluation says it all, very poor memory or you lack concentration
45-59	SIMPLY POOR	Your memory is poor, you'll need hard training to try and improve it
60-74	JUST AVERAGE	Average memory but still not a lost cause, train regularly to improve it
75-89	QUITE GOOD	Decent memory, a little training will get you to the next level in no time
90-104	REALLY GOOD	Pretty good memory, playing this game regularly will do wonders
105-119	TRULY SUPERB	Quite impressive, either you're a natural or you've trained really hard
≥ 120	PHOTOGRAPHIC	Wow, eidetic memory, you're a natural for sure, hats off to you

In order to generate the pseudo-random integers for the game, and as the HP-41C lacks a built-in RNG (*Random Number Generator*), the program implements a simple one but you must first of all store a *seed* (some nonnegative number of your choice, see **Note 1**) in register R_{01} before running the program. This needs to be done just once per session, no matter how many games you play afterwards.

2. Program Listing

```

01 ◆LBL "MEMORY" 31 PROMPT      61 FRC          91 GTO 14 ▶      121 ◆LBL 02
02 "HERE WE GO" 32 X<=0?      62 ST- 02       92 RCL 00        122 "VERY POOR"
03 AVIEW        33 GTO 09 ▶      63 RCL 03       93 1             123 PROMPT
04 PSE         34 ENTER↑     64 FRC          94 ST+ 04        124 ◆LBL 03
05 FIX 0       35 LOG         65 ST- 03       95 X<=Y?        125 "SIMPLY POOR"
06 CF 29       36 INT         66 X=Y?         96 ST- 00        126 PROMPT
07 CLX         37 RCL 00      67 GTO 11 ▶     97 GTO 10 ▶     127 ◆LBL 04
08 STO 00      38 -           68 SF 00        98 ◆LBL 14       128 "JUST AVERAGE"
09 STO 04      39 10↑X        69 1            99 BEEP         129 PROMPT
10 STO 05      40 /           70 ST- 05      100 "YOU GOT "   130 ◆LBL 05
11 ◆LBL 10     41 INT         71 GTO 11 ▶    101 ARCL 05     131 "QUITE GOOD"
12 CLD         42 STO 03      72 ◆LBL 12     102 +" RIGHT"   132 PROMPT
13 RCL 01      43 RCL 02      73 "RIGHT"     103 AVIEW        133 ◆LBL 06
14 R-D         44 X=Y?        74 AVIEW       104 PSE          134 "REALLY GOOD"
15 FRC         45 GTO 12 ▶    75 ◆LBL 15     105 "YOUR MEMORY IS" 135 PROMPT
16 STO 01      46 "WRONG: "   76 1           106 AVIEW        136 ◆LBL 07
17 9           47 RCL 04      77 ST+ 05     107 PSE          137 "TRULY SUPERB"
18 *           48 1           78 RCL 00     108 8            138 PROMPT
19 1           49 +           79 ST+ 05     109 RCL 05      139 ◆LBL 08
20 +           50 ARCL X     80 FS?C 00    110 15           140 "PHOTOGRAPHIC"
21 RCL 00      51 AVIEW       81 GTO 13 ▶    111 /           141 PROMPT
22 10↑X        52 ◆LBL 11     82 9           112 X>Y?        142 END
23 *           53 RCL 03      83 X>Y?       113 X<>Y
24 INT         54 RCL 02      84 ISG 00     114 GTO IND X ▶
25 STO 02      55 X=Y?        85 GTO 10 ▶    115 ◆LBL 00
26 PSE         56 GTO 15 ▶    86 GTO 10 ▶    116 "NEAR AMNESIA"
27 PSE         57 10          87 ◆LBL 13     117 PROMPT
28 CLX         58 ST/ 02      88 2           118 ◆LBL 01
29 ◆LBL 09     59 ST/ 03      89 RCL 04     119 "SO PATHETIC"
30 "NUMBER?"   60 RCL 02      90 X=Y?       120 PROMPT

```

Remarks:

- 142 steps, requires at least SIZE 006, will fit in a basic HP-41C with no memory modules.
- uses flag 00, clears flag 29 and sets display mode FIX 0.
- to get * press **X**, to get / press **÷**, to get "text" press **ALPHA**, to get +"text" use Append.
- the symbols **◆** and **▶** are purely cosmetic, to visually indicate branching, don't try to key them in.

3. Usage Instructions

See the following example to understand how to use the program.

4. Example

The following example can be useful to check that the program is correctly entered and to understand its usage:

Using 0.1980 as a seed, try and test your memory by running the program, like this:

```
0.1980 STO 01 (store the seed for the RNG, do it just once per session, no matter how many games you play afterwards)
```

XEQ	"MEMORY"	→	HERE WE GO	→	4	→	NUMBER?	(first try, just one digit to remember)
4	R/S	→	RIGHT	→	76	→	NUMBER?	(Ok, one digit longer than before)
76	R/S	→	RIGHT	→	566	→	NUMBER?	(Ok, one digit longer than before)
566	R/S	→	RIGHT	→	7212	→	NUMBER?	(Ok, one digit longer than before)
7212	R/S	→	RIGHT	→	59304	→	NUMBER?	(Ok, one digit longer than before)
59604	R/S	→	WRONG: 1	→	4493	→	NUMBER?	(1 st oops, one digit shorter than before)
4496	R/S	→	WRONG: 2	→	316	→	NUMBER?	(2 nd oops, one digit shorter than before)
316	R/S	→	RIGHT	→	7861	→	NUMBER?	(Ok, one digit longer than before)
7831	R/S	→	WRONG: 3	→	YOU GOT 23 RIGHT			(3 rd and last failure, game ends)
				→	YOUR MEMORY IS			
				→	SO PATHETIC			

Well, it didn't go that well this time but this was just an example.

Why don't you give it a try for real and see how you fare ?

Notes

1. Don't use 0 or negative seeds and also avoid **PI** and its multiples or fractions, as well as very large numbers.
2. Once the evaluation message has appeared the run is *ended*, don't press **R/S** to play another game but **XEQ** "MEMORY" again instead.
3. If you'd prefer having the number stay for longer than 2 seconds, insert as many PSE steps as desired after 27 PSE for an easier game.
4. If you'd rather get *better* evaluations with the *same* difficulty level, you may change the hard-coded *calibration constant* **15** at *step 110* to a smaller value, say **10**. Actually you may *fine tune* the evaluation to better suit your feelings about what would be *fair* by changing the **15** to larger or smaller values. If you find the game too easy (too good evals all the time) change it to **20** or higher. If too hard (always bad evals), change it to **10** or less. For instance, with the default values I can go on till I tire, which is around a score of 388, so I use **25**.
5. Practice makes perfect: try some sessions each day, concentrate *hard*, and you might find your memory evaluations increasing over time. Once training improves your memory (you mostly get *PHOTOGRAPHIC*), change the **15** to **20** and/or *delete* 27 PSE to keep improving.
6. This training side of the game can actually be *very useful* to help keep the memory from degrading or even to noticeably improve it.

Copyrights

Copyright for this paper and its contents is retained by the author. Permission to use it for non-profit purposes is granted as long as the contents aren't modified in any way and the copyright is acknowledged.